

NAG Fortran Library Routine Document

F08KAF (DGELSS)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08KAF (DGELSS) computes the minimum norm solution to a real linear least squares problem

$$\min_x \|b - Ax\|_2.$$

2 Specification

```
SUBROUTINE F08KAF (M, N, NRHS, A, LDA, B, LDB, S, RCOND, RANK, WORK,
1                  LWORK, INFO)
    INTEGER          M, N, NRHS, LDA, LDB, RANK, LWORK, INFO
    double precision A(LDA,*), B(LDB,*), S(*), RCOND, WORK(*)
```

The routine may be called by its LAPACK name ***dgelss***.

3 Description

F08KAF (DGELSS) uses the singular value decomposition (SVD) of A , where A is an m by n matrix which may be rank-deficient.

Several right-hand side vectors b and solution vectors x can be handled in a single call; they are stored as the columns of the m by r right-hand side matrix B and the n by r solution matrix X .

The effective rank of A is determined by treating as zero those singular values which are less than RCOND times the largest singular value.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: <http://www.netlib.org/lapack/lug>

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

- 1: M – INTEGER *Input*
On entry: m , the number of rows of the matrix A .
Constraint: $M \geq 0$.
- 2: N – INTEGER *Input*
On entry: n , the number of columns of the matrix A .
Constraint: $N \geq 0$.
- 3: NRHS – INTEGER *Input*
On entry: r , the number of right-hand sides, i.e., the number of columns of the matrices B and X .
Constraint: $NRHS \geq 0$.

- 4: A(LDA,*) – **double precision** array *Input/Output*
Note: the second dimension of the array A must be at least $\max(1, N)$.
On entry: the m by n matrix A .
On exit: the first $\min(m, n)$ rows of A are overwritten with its right singular vectors, stored rowwise.
- 5: LDA – INTEGER *Input*
On entry: the first dimension of the array A as declared in the (sub)program from which F08KAF (DGELSS) is called.
Constraint: $LDA \geq \max(1, M)$.
- 6: B(LDB,*) – **double precision** array *Input/Output*
Note: the second dimension of the array B must be at least $\max(1, NRHS)$.
On entry: the m by r right-hand side matrix B .
On exit: is overwritten by the n by r solution matrix X . If $m \geq n$ and $RANK = n$, the residual sum-of-squares for the solution in the i th column is given by the sum of squares of elements $n + 1, \dots, m$ in that column.
- 7: LDB – INTEGER *Input*
On entry: the first dimension of the array B as declared in the (sub)program from which F08KAF (DGELSS) is called.
Constraint: $LDB \geq \max(1, \max(M, N))$.
- 8: S(*) – **double precision** array *Output*
Note: the dimension of the array S must be at least $\max(1, \min(M, N))$.
On exit: the singular values of A in decreasing order.
- 9: RCOND – **double precision** *Input*
On entry: used to determine the effective rank of A . Singular values $S(i) \leq RCOND \times S(1)$ are treated as zero.
If $RCOND < 0$, **machine precision** is used instead.
- 10: RANK – INTEGER *Output*
On exit: the effective rank of A , i.e., the number of singular values which are greater than $RCOND \times S(1)$.
- 11: WORK(*) – **double precision** array *Workspace*
Note: the dimension of the array WORK must be at least $\max(1, LWORK)$.
On exit: if $INFO = 0$, $WORK(1)$ returns the optimal LWORK.
- 12: LWORK – INTEGER *Input*
On entry: the dimension of the array WORK as declared in the (sub)program from which F08KAF (DGELSS) is called.
For good performance, LWORK should generally be larger. Consider increasing LWORK by at least $nb \times \min(M, N)$, where nb is the optimal block size.

If $LWORK = -1$, a workspace query is assumed; the routine only calculates the optimal size of the $WORK$ array, returns this value as the first entry of the $WORK$ array, and no error message related to $LWORK$ is issued.

Constraint: $LWORK \geq 1$, and also

$LWORK \geq 3 \times \min(M, N) + \max(2 \times \min(M, N), \max(M, N), NRHS)$.

13: INFO – INTEGER

Output

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO = $-i$, the i th argument had an illegal value.

INFO > 0

The algorithm for computing the SVD failed to converge; if INFO = i , i off-diagonal elements of an intermediate bidiagonal form did not converge to zero.

7 Accuracy

See Section 4.5 of Anderson *et al.* (1999) for details.

8 Further Comments

The complex analogue of this routine is F08KNF (ZGELSS).

9 Example

To solve the linear least squares problem

$$\min_x \|b - Ax\|_2$$

for the solution, x , of minimum norm, where

$$A = \begin{pmatrix} -0.09 & 0.14 & -0.46 & 0.68 & 1.29 \\ -1.56 & 0.20 & 0.29 & 1.09 & 0.51 \\ -1.48 & -0.43 & 0.89 & -0.71 & -0.96 \\ -1.09 & 0.84 & 0.77 & 2.11 & -1.27 \\ 0.08 & 0.55 & -1.13 & 0.14 & 1.74 \\ -1.59 & -0.72 & 1.06 & 1.24 & 0.34 \end{pmatrix}$$

and

$$b = \begin{pmatrix} 7.4 \\ 4.2 \\ -8.3 \\ 1.8 \\ 8.6 \\ 2.1 \end{pmatrix}.$$

A tolerance of 0.01 is used to determine the effective rank of A .

Note that the block size (NB) of 64 assumed in this example is not realistic for such a small problem, but should be suitable for large problems.

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      F08KAF Example Program Text
*      Mark 21 Release. NAG Copyright 2004.
*      .. Parameters ..
INTEGER      NIN, NOUT
PARAMETER    (NIN=5,NOUT=6)
INTEGER      MMAX, NB, NMAX
PARAMETER    (MMAX=16,NB=64,NMAX=8)
INTEGER      LDA, LWORK
PARAMETER    (LDA=MMAX,LWORK=3*NMAX+NB*(MMAX+NMAX))
*      .. Local Scalars ..
DOUBLE PRECISION RCOND, RNORM
INTEGER      I, INFO, J, M, N, RANK
*      .. Local Arrays ..
DOUBLE PRECISION A(LDA,NMAX), B(MMAX), S(NMAX), WORK(LWORK)
*      .. External Functions ..
DOUBLE PRECISION DNRM2
EXTERNAL      DNRM2
*      .. External Subroutines ..
EXTERNAL      DGELSS
*      .. Executable Statements ..
WRITE (NOUT,*) 'F08KAF Example Program Results'
WRITE (NOUT,*)
*      Skip heading in data file
READ (NIN,*)
READ (NIN,*) M, N
IF (M.LE.MMAX .AND. N.LE.NMAX .AND. M.GE.N) THEN

*
*      Read A and B from data file
*
      READ (NIN,*) ((A(I,J),J=1,N),I=1,M)
      READ (NIN,*) (B(I),I=1,M)

*
*      Choose RCOND to reflect the relative accuracy of the input data
*
      RCOND = 0.01D0

*
*      Solve the least squares problem min( norm2(b - Ax) ) for the x
*      of minimum norm.
*
      CALL DGELSS(M,N,1,A,LDA,B,M,S,RCOND,RANK,WORK,LWORK,INFO)

*
      IF (INFO.EQ.0) THEN

*
*      Print solution
*
        WRITE (NOUT,*) 'Least squares solution'
        WRITE (NOUT,99999) (B(I),I=1,N)

*
*      Print the effective rank of A
*
        WRITE (NOUT,*)
        WRITE (NOUT,*) 'Tolerance used to estimate the rank of A'
        WRITE (NOUT,99998) RCOND
        WRITE (NOUT,*) 'Estimated rank of A'
        WRITE (NOUT,99997) RANK

*
*      Print singular values of A
*
        WRITE (NOUT,*)
        WRITE (NOUT,*) 'Singular values of A'
        WRITE (NOUT,99999) (S(I),I=1,N)

*
*      Compute and print estimate of the square root of the
*      residual sum of squares
*

```

```

      IF (RANK.EQ.N) THEN
        RNORM = DNRM2(M-N,B(N+1),1)
        WRITE (NOUT,*)
        WRITE (NOUT,*)
+      'Square root of the residual sum of squares'
        WRITE (NOUT,99998) RNORM
      END IF
    ELSE
      WRITE (NOUT,*) 'The SVD algorithm failed to converge'
    END IF
  ELSE
    WRITE (NOUT,*) 'MMAX and/or NMAX too small, and/or M.LT.N'
  END IF
  STOP
*
99999 FORMAT (1X,7F11.4)
99998 FORMAT (3X,1P,E11.2)
99997 FORMAT (1X,I6)
END

```

9.2 Program Data

F08KAF Example Program Data

6	5	:Values of M and N			
-0.09	0.14	-0.46	0.68	1.29	
-1.56	0.20	0.29	1.09	0.51	
-1.48	-0.43	0.89	-0.71	-0.96	
-1.09	0.84	0.77	2.11	-1.27	
0.08	0.55	-1.13	0.14	1.74	
-1.59	-0.72	1.06	1.24	0.34	:End of matrix A
7.4					
4.2					
-8.3					
1.8					
8.6					
2.1					:End of vector b

9.3 Program Results

F08KAF Example Program Results

Least squares solution

0.6344	0.9699	-1.4403	3.3678	3.3992
--------	--------	---------	--------	--------

Tolerance used to estimate the rank of A

1.00E-02

Estimated rank of A

4

Singular values of A

3.9997	2.9962	2.0001	0.9988	0.0025
--------	--------	--------	--------	--------
